

Global Comparative Statics in General Equilibrium: model building from theoretical foundations

James R. Markusen
Department of Economics
University of Colorado, Boulder

Abstract

International trade economists made seminal contributions to general equilibrium theory, moving away from an emphasis on existence of equilibrium to algebraic formulations which enabled us to characterize key relationships between parameters and variables, such as that between tariffs and domestic factor prices and welfare. But the analysis remained limited in value for policy evaluation: the analysis was local, it provided only qualitative results, it was limited to very small models, and strictly interior solutions had to be assumed. The contribution of this paper is pedagogic and methodological, providing a primer for those wishing to do or teach general-equilibrium counterfactuals on computable general-equilibrium (CGE) or structural econometric models. I show how the tools from early local comparative statics analyses can be generalized via the use of Shepard's lemma, duality, complementarity and the Karush-Kuhn-Tucker theorem into a global, quantitative analysis of large changes in high-dimension models which also allows for regime changes and corner solutions. I then show how the resulting non-linear complementarity problem directly translates into a numerical model using GAMS (general algebraic modeling system). The paper concludes with two examples: (a) comparison of a tax versus a real trade/transactions cost, (b) comparison of a tax versus a quantitative restriction such as a quota or license.

Keywords: duality, complementarity, KKT conditions, global comparative statics, corner solutions, GAMS

JEJ codes: F23, D50, D58

This paper is a re-orientated and much expanded version of a paper prepared for a special issue of the *International Journal of Economic Theory* in honor of Ronald W. Jones (Markusen 2021). For a great deal of further information on modeling complete with the GAMS code for many examples, please see my website <http://spot.colorado.edu/~markusen>.

The author thanks Alan Woodland and Thomas Rutherford for suggestions and perspectives. Not only for helpful comments and suggestions on this paper, but for making essential contributions to the theory underlying the paper. Alan Woodland for major contributions to the development of duality theory and Thomas Rutherford for developing and implementing the complementarity formulation of general equilibrium models. In doing so, Rutherford simplified and standardized applied general-equilibrium analysis, allowing economists to concentrate on economics and policy analysis rather than on coding.

May 17, 2021

1. Introduction

Simulation analysis of general equilibrium models used to be the territory of the field of applied general-equilibrium (also called computable general equilibrium) and members of this group were largely disjoint from international trade theorists and indeed empiricists using the tool kit of econometrics. I think it is fair to say that there was even some hostility among these groups. I regret not keeping some of the referees' reports on my early papers using numerical simulation as a theory tool in models far too complex for traditional paper-and-pencil analytical methods. But reading journals and attending conferences today makes it clear that, slowly but surely, simulation analysis has moved into the mainstream of international economics and other fields. My casual empiricism is that some authors are creating simulations of analytical models on an ad hoc basis that do not allow transparency and verifiability (e.g., is the model correctly computing general-equilibrium comparative statics). Many applied general-equilibrium (AGE) modelers might also benefit from a clearer understanding of the theoretical foundations needed for a good model.

The purpose of this paper is to explore the theoretical roots of a proper and robust general-equilibrium model. I then show how the theory-based framework easily translates into a computable numerical simulation model. As such, the paper is pedagogic and methodological. It makes no claim to offer original theoretical insights or results, but rather shows how familiar existing tools and well-known mathematical results combine to produce a simple, clear and consistent modeling framework. A short appendix to the paper outlines several other modeling frameworks and their origins, but the paper is not concerned with evaluating alternative software and solution algorithms. While the motivation and focus here is on general equilibrium, the tools and concepts I discuss are applicable to a huge range of problems in microeconomics.

International trade economists' early contributions in formalizing our basic general-equilibrium model were a seminal contribution that reworked how we think about general equilibrium (GE). Early articles such as Jones (1956, 1965) remained standard readings for graduate students for decades. Jones and others moved us away from focusing on issues of existence, uniqueness and stability of equilibrium to a more useful concentration on the actual properties of GE that we might be interested in as applied micro economists in fields like international trade and public economics. How does, for example, a trade tariff affect the internal distribution of income in an economy? GE analysis focusing on existence of equilibrium and fixed-point algorithms is of little practical value for applied questions.

One particularly important development was to provide early versions of what we would now call duality analysis. I will have more to say on this below, but basically it moved us from looking at production and utility functions to using cost functions which embody optimizing behavior at the level of firms and households. Then a general-equilibrium model can be built up, embodying optimization by individual agents in the equations and inequalities of the model.

These early contributions were path breaking and remain important today. But there are of course limitations to the usefulness of the approach. Without in any way disparaging the importance of analytical contributions, some of these are as follows. First, the analysis is of small or "local" (differential) changes and cannot easily be extended to large "global" changes.

Second, the results are qualitative, giving signs but not magnitudes of effects in comparative-statics experiments. Third, the techniques cannot give even sign predictions past very simple cases such as a two-good, two-factor economy. Fourth, the comparative-statics methods cannot easily handle corner solutions in which parameter changes lead countries to change the set of goods they actually produce, switch technologies used to produce some goods, or cause changes in which trade links are active and inactive.

My objective is thus to indicate how local analysis can be extended to a global analysis, which allows for the quantitative evaluation of large parameter changes and permits changes in trade and production specialization patterns (e.g., which production sectors and which trade links are active or inactive). I will show how this global analysis is rigorously built up from several key results from mathematics and economic theory. These generalizations are hugely important in the evaluation of large policy changes such as BREXIT, or economic shocks such as covid-19 where qualitative local analyses of small changes are of no practical value. The ability to handle corner solutions and regime shifts is similarly crucial to analyses of global value chains, expansion of trade at the extensive margin, trade diversion and multinational firm location decisions.

I then present two specific cases which illustrate important how modeling decision may create substantially different implications when parameter changes are large. In the first, I compare a tax with lump-sum redistribution to a real-resource-using transaction or trade cost. In the second, I compare the effect of a tax and a quota, which are calibrated to be equivalent initially, but which lead to large differences when parameter changes are large. Both of these situations are, I believe, quite common elements of AGE models used for policy analysis.

2. From local to global analysis¹

Let X and Y be two goods with prices p_x and p_y . There are two factors of production in fixed supply, denoted L and K with prices w and r respectively. Utility of a representative consumer is denoted U and income is denoted I .

We begin by drawing on traditions that were partly implicit in Jones' work, and developed further later on in books explicitly focusing on duality such as Takayama and Woodland (1980), Woodland (1980, 1982) and Dixit and Norman (1980). The trick is to first move from production and utility functions to cost functions, with those cost functions embodying not just technologies but also the optimizing behavior of individual firms and households. These can be termed "value functions": the endogenous choices of inputs and outputs by firms and households are solved for by standard optimization methods and then inserted back into cost equations to get the minimum cost of producing goods or utility as a

¹In order to avoid confusion with terminology in international trade, I note again that "local" refers to small (differential) changes, not to a closed economy. "Global" refers to large, discrete changes in parameters, not to multi-country analysis.

function of parameters or variables that the agent views as exogenous (e.g., consumers take prices as exogenous, but they are endogenous variables in general equilibrium).

In our 2x2 case outlined above, we can derive four value functions by standard optimization methods which use the Karush-Kuhn-Tucker theorem (Karush 1939, Kuhn and Tucker 1951) as the underlying methodology. Making the usual assumption that production and utility exhibit constant returns to scale, cost functions, the expenditure function and the indirect utility function are separable in outputs, utility and income. These four are given as follows.

$$C_x(w, r, X) = c_x(w, r)X \quad \text{minimum total cost of } X \quad (1)$$

$$C_y(w, r, Y) = c_y(w, r)Y \quad \text{minimum total cost of } Y \quad (2)$$

$$E(p_x, p_y, U) = e(p_x, p_y)U \quad \text{minimum cost of utility level } U \quad (3)$$

$$V(p_x, p_y, I) = v(p_x, p_y)I \quad \text{maximum (indirect) utility level } U \quad (40)$$

Though it is obvious and well-known to most readers, note that the expenditure function is just a cost function under a different name: it gives the minimum cost at existing commodity prices needed to purchase one unit of utility. With this constant returns / separability assumption, unit cost, expenditure and utility functions are simply

$$c_x = c_x(w, r), \quad c_y = c_y(w, r) \quad \text{unit cost functions for } X \text{ and } Y \quad (5)$$

$$e = e(p_x, p_y) \quad \text{unit cost (expenditure) function} \quad (6)$$

$$v = v(p_x, p_y) \quad \text{indirect utility per unit of income} \quad (7)$$

The next crucial step is also provided by theory. Shephard's lemma and Roy's identity, which follow from the envelope theorem, implies that the partial derivatives of these value functions give us the optimal choices of inputs and outputs given prices for goods and factors.² Let a_{ij} denote the optimal, or cost minimizing, amount of factor i needed to produce one unit of good j . Cost minimization implies that the a_{ij} are themselves functions of w and r . Let h_x denote the Hicksian demand for X per unit of utility and let d_x denote the Marshallian demand for X per unit of income.

$$\frac{\partial c_x}{\partial w} = a_{lx} \quad \text{optimal amount of labor per unit of } X \text{ output} \quad (8)$$

$$\frac{\partial c_x}{\partial r} = a_{kx} \quad \text{optimal amount of capital per unit of } X \text{ output} \quad (9)$$

²Shepard (1953), Roy (1947). The envelop theorem is generally credited to Auspitz and Lieben (1889), who provide a number of examples but do use the term envelop theorem. See Schmidt (2004). See also Hotelling (1932) for a related result on profit functions, which is not of use in our type of model because constant returns to scale does not satisfy strict concavity of the profit function.

$$\frac{\partial e}{\partial p_x} = h_x \quad \text{consumer's demand for } X \text{ per unit of utility (Hicksian)} \quad (10)$$

with corresponding equations for the sector Y cost function and Y demand. Roy's identity gives The Marshallian demand for X : demand as a function of prices and income.

$$-\frac{\partial V}{\partial p_x} / \frac{\partial V}{\partial I} = -\frac{\partial v}{\partial p_x} \frac{I}{v} = X = d_x I \quad (11)$$

where d_x is the demand for X per unit of income I . This gives

$$-\frac{\partial v}{\partial p_x} / v = d_x \quad \text{consumer demand for } X \text{ per unit of income (Marshallian)} \quad (12)$$

The modern approach to AGE modeling uses these tools as the building blocks for a global analysis. In addition to having the ability to evaluate large changes such as large-scale trade liberalization or tax reform, the newer approach permits corner solutions in which some production activities or trade links can switch from inactive to active and vice versa as a consequence of parameter changes. This requires us to detour a bit into complementarity, a concept that follows directly from the Karush-Kuhn-Tucker (KKT) theorem.

3. General equilibrium and complementarity

Equilibrium is modeled as a set of weak inequalities each with a complementary non-negative variable. Pricing inequalities are written as unit cost greater than or equal to price, with output of that activity (industry, trade flow, etc.) being the complementary variable. If the activity is unprofitable in equilibrium (strict inequality), it is not used and the complementary output variable is zero. Market clearing inequalities are strictly speaking not KKT optimization conditions, but rather equilibrium conditions. Yet they can be handled in a way closely equivalent to KKT conditions. A market clearing inequality is written as supply greater than or equal to demand, with price being the complementary variable: if supply exceeds demand in equilibrium, then it is a free good. The KKT theorem introduces added "slack" or complementary variables so that the weak inequalities become equations, which then allows solver algorithms to use iterative methods to solve the system of equations.

The strong microeconomic foundations of duality and complementarity via the KKT theorem led one group of modelers to formulate GE models as a sequence of complementarity problems at the level of industries and households.³ Notable in this development were contributions by Mathiesen (1985) and Rutherford (1985). Rutherford's MPS/GE (mathematical

³An appendix will provide a short description of each of several approaches to formulating AGE models. As noted above, I will not attempt to evaluate these against one another.

programming for general equilibrium) allowed for the easy calibration and implementation of the complementarity approach. An early example of this is Harrison, Wooton, Rutherford (1989).

Now let's look at an actual implementation in which optimization is embodied at the level of industries and households. We stick with our two-good, two-factor model from above, with a representative consumer and a closed economy. First, let's use the Marshallian demand formulation for the consumer. A seven equation, seven variable model is as follows. There are two pricing equations, four market-clearing equations, and one income-balance equation.

$$c_x(w, r) \geq p_x \quad \perp \quad X \quad \text{pricing (zero-profit) inequalities} \quad (13)$$

$$c_y(w, r) \geq p_y \quad \perp \quad Y \quad (14)$$

$$L \geq a_{lx}X + a_{ly}Y \quad \perp \quad w \quad (15)$$

$$K \geq a_{kx}X + a_{ky}Y \quad \perp \quad r \quad (16)$$

market-clearing inequalities

$$X \geq d_x I \quad \perp \quad p_x \quad (17)$$

$$Y \geq d_y I \quad \perp \quad p_y \quad (18)$$

$$I = wL + rK \quad \perp \quad I \quad \text{Income balance} \quad (19)$$

Note that the first four weak inequalities are Jones' (1965) equations: if the prices of X and Y are fixed by world markets (small open economy assumption), then these four can be solved on their own for the four complementary variables.

After solving this model, the utility and price index for the representative consumer can be calculated. All alternative procedure, especially useful when there are multiple household types or countries is to use a Hicksian formulation. This treats utility as if it were a produced good: commodities are inputs into the production of a utility good, and the expenditure function is the minimum cost of producing one unit. There is also a (virtual) market for the utility good, with a market clearing equation and complementary variable the price of a unit of utility. This is what we generally label as the consumer price index. This model is a little more complicated, but it computes utility and the price index for each household type or country as part of the solution. Denote U as utility and p_u as the (consumer) price index, the cost of one unit of utility. Our extended Hicksian model is given by nine weak inequalities in nine unknowns.

$$e(p_x, p_y) \geq p_u \quad \perp \quad U \quad (20)$$

$$c_x(w, r) \geq p_x \quad \perp \quad X \quad \text{pricing (zero profit) inequalities} \quad (21)$$

$$c_y(w, r) \geq p_y \quad \perp \quad Y \quad (22)$$

$$L \geq a_{lx}X + a_{ly}Y \quad \perp \quad w \quad (23)$$

$$K \geq a_{kx}X + a_{ky}Y \quad \perp \quad r \quad (24)$$

$$X \geq h_x U \quad \perp \quad p_x \quad \text{market-clearing inequalities} \quad (25)$$

$$Y \geq h_y U \quad \perp \quad p_y \quad (26)$$

$$U \geq I/p_u \quad \perp \quad p_u \quad (27)$$

$$I = wL + rK \quad \perp \quad I \quad \text{income balance} \quad (28)$$

Following earlier comments, the strength of this approach is that it computes equilibria for large changes in parameters and high-dimension models, it will give quantitative results, and it allows for corner solutions in which some variables switch from slack (equal to zero) to positive or vice versa. However, there are also some limitations. First, an implementation requires explicit functional forms for production and utility. Furthermore, quantitative analysis requires that numerical parameter values must be chosen for those functional forms. I and others using numerical models as a theory tool acknowledge this tradeoff, but note that insisting on analytical models only often requires the modeler to simplify the model so much that the interesting parts of the problem are discarded.⁴

4. A numerical implementation

We can now look at an actual implementation of our model using the Hicksian formulation. As just noted, it has the advantage of computing utility and the price index as part of the solution, a big advantage in multi-country, multi-household economies. We use Cobb-Douglas functions for the three activities so that it is simple and straightforward for readers to see exactly where the equations and inequalities are coming from. There is no attempt to base parameter values on any real-world equivalents, rather they are chosen to provide maximum transparency. Goods in the utility function get equal shares of 0.5. The representative consumer's utility function and the implied expenditure function are given as follows.

$$U(X, Y) = 2 * X^{0.5} Y^{0.5} \quad \Rightarrow \quad e(p_x, p_y) = p_x^{0.5} p_y^{0.5} = p_u \quad (29)$$

⁴The complementarity approach is adaptable to very complex economies. For models with increasing returns to scale, imperfect competition, endogenous markups, and endogenous firm location decisions, see Markusen (2002). Examples of these techniques used in very theory models, see Markusen and Venables (2007) (29,000 non-linear inequalities and unknowns) and Markusen (2013) (36,000 non-linear inequalities and unknowns).

Using a textbook Lagrangean optimization formulation of KKT, Marshallian and Hicksian unit demand functions are as follows.

$$d_x = 0.5/p_x \quad d_y = 0.5/p_y \quad (30)$$

$$h_x = 0.5 * p_x^{-0.5} p_y^{0.5} = 0.5 * (p_u/p_x) \quad h_y = 0.5 * p_x^{0.5} p_y^{-0.5} = 0.5 * (p_u/p_y) \quad (31)$$

In order to verify the relationship between utility and income as in (27) above, multiply the Marshallian unit demands in (30) by X and Y respectively and insert them into (29).

$$U(X, Y) = 2 * (0.5 * I/p_x)^{0.5} * (0.5 * I/p_y)^{0.5} = I * p_x^{-0.5} * p_y^{-0.5} = I/p_u \quad (32)$$

X is (arbitrarily) capital intensive: a capital share of 0.75, a labor share of 0.25.

$$c_x(w, r) = w^{0.25} r^{0.75} \quad (33)$$

$$a_{lx} = 0.25 * w^{-0.75} r^{0.75} = 0.25 * (r/w)^{0.75} \quad (34)$$

$$a_{kx} = 0.75 * w^{0.25} r^{-0.25} = 0.75 * (w/r)^{0.25} \quad (35)$$

Y is labor intensive with the opposite ordering of shares.

$$c_y(w, r) = w^{0.75} r^{0.25} \quad (36)$$

$$a_{ly} = 0.75 * w^{-0.25} r^{0.25} = 0.75 * (r/w)^{0.25} \quad (37)$$

$$a_{ky} = 0.25 * w^{0.75} r^{-0.75} = 0.25 * (w/r)^{0.75} \quad (38)$$

Let \bar{L} ($LBAR$) and \bar{K} ($KBAR$) denote economy's fixed endowments of labor and capital. We now have a complete numerical model with only two parameters to be chosen which are the endowment quantities. Here then is the implemented model.

$$p_x^{0.5} p_y^{0.5} \geq p_u \quad \perp \quad U \quad (39)$$

$$w^{0.25} r^{0.75} \geq p_x \quad \perp \quad X \quad (40)$$

$$w^{0.75} r^{0.25} \geq p_y \quad \perp \quad Y \quad (41)$$

$$LBAR \geq 0.25 * (r/w)^{0.75} X + 0.75 * (r/w)^{0.25} Y \quad \perp \quad w \quad (42)$$

$$KBAR \geq 0.75 * (w/r)^{0.25} X + 0.25 * (w/r)^{0.75} Y \quad \perp \quad r \quad (43)$$

$$X \geq 0.5 * (p_u/p_x) U \quad \perp \quad p_x \quad (44)$$

$$Y \geq 0.5 * (p_u/p_y) U \quad \perp \quad p_y \quad (45)$$

$$U \geq I/p_u \quad \perp \quad p_u \quad (46)$$

$$I = w * LBAR + r * KBAR \quad \perp \quad I \quad (47)$$

While any values of $LBAR$ and $KBAR$ will produce a solution, it is a good practice to start with a calibrated solution as a check on the modeler's consistency. This is generally referred to as the "replication check": running the model should yield the initial calibrated values as a solution, otherwise something is wrong. If we choose $LBAR = 100$ and $KBAR = 100$, then due to the symmetry in production and consumption shares, we should get a solutions in which $X = Y = 100$, $I = 200$, and all prices equal one.

The principal objective of this paper is pedagogic: demonstrate how to move from a traditional algebraic model used for local comparative-statics experiments to more robust and useful global comparative statics. According, I will show an actual numerical model. The code is available from me on request. I hope that by showing the code, I can help aspiring modelers to get a big head start in seeing that global general-equilibrium analysis need not be a black box.

GAMS (general algebraic modeling system) is the modeling language and solvers that I will present here. GAMS is an algebraic language and thus it is intuitive and relatively easy to master. Equations are written exactly as they are done here in the text, and there are no weird symbols or characters that need to be memorized in order to do straightforward things. In addition, the solvers in GAMS are constructed on the basis of theory (KKT), particularly the MCP (mixed complementary problem) solver PATH which uses a generalization of Newton's method.⁵

A quick note: as a consequence of Walras Law, there is an indeterminacy of the price level in the model, so one price is chosen as numeraire and fixed at one. The complementary equation is then automatically dropped by GAMS from the model. The price of utility $U(p_u)$ is chosen as numeraire and its price fixed at one.⁶ This has the advantage that other prices, such as the wage rate, are "real" prices, measured relative to the consumer price index.

⁵A referee nicely provided this model written in GEMPACK code, which I will comment on very briefly in a short appendix. This is found at <http://www.copsmodels.com/archivep.htm#tpmh0191>

⁶In GAMS, PY.FX denotes fixing the variable PY; then the equation complementary to that variable is automatically dropped from the model. PY.L is the notation for setting the initial value or level of variable PY, but that variable is not held fixed. Setting initial values of variables is important for the solver to solve the model and solve efficiently in all non-linear problems.

\$TITLE: JGEA1 basic model,

\$ONTEXT

This model is a closed economy version of the classic Heckscher-Ohlin model: two goods and two factors, one consumer

Consumer endowed with labor and capital, which are used to produce X and Y
 X and Y are treated as though they combine to produce utility U
 Consumer buys U with the income from labor and capital

The following table shows the model calibration

Row sum = zero indicates market clearing

Column sum = zero indicates zero profits for X and Y or income balance

Markets	Production Sectors			Consumers	
	X	Y	U	I	
PX	100		-100		
PY		100	-100		
W	-25	-75			100
R	-75	-25			100
PU			200		-200

\$OFFTEXT

PARAMETERS

LBAR labor endowment
 KBAR capital endowment;

LBAR = 100;

KBAR = 100;

NONNEGATIVE VARIABLES

U activity level for utility or welfare
 X activity level for X production
 Y activity level for Y production

 PX price of good X
 PY price of good Y
 W price of labor
 R price of capital
 PU price of welfare (expenditure function)

 I income of the representative consumer;

EQUATIONS

PRF_U zero profit for welfare
 PRF_X zero profit for sector X
 PRF_Y zero profit for sector Y

 MKT_X supply-demand balance for commodity X
 MKT_Y supply-demand balance for commodity Y
 MKT_L supply-demand balance for primary factor L
 MKT_K supply-demand balance for primary factor K
 MKT_U supply-demand balance for welfare

 INC_I income balance;

```

*      List equation names followed by double dot ..
*      then write out weak inequalities in greater than or equal to format

*      Zero profit inequalities

PRF_X..      W**0.25 * R**0.75 =G= PX;
PRF_Y..      W**0.75 * R**0.25 =G= PY;
PRF_U..      PX**0.5 * PY**0.5 =G= PU;

*      Market clearance inequalities

MKT_L..      LBAR =G= 0.25*(R/W)**0.75*X +
              0.75*(R/W)**0.25*Y;

MKT_K..      KBAR =G= 0.75*(W/R)**0.25*X +
              0.25*(W/R)**0.75*Y;

MKT_X..      X      =G= 0.5*U*PU/PX;
MKT_Y..      Y      =G= 0.5*U*PU/PY;
MKT_U..      U      =G= I/PU;

*      Income balance equation

INC_I..      I =E= LBAR*W + KBAR*R;

*      Declare a model name, list equation names followed by dot and name
*      of complementary variable

MODEL CLOSED /PRF_U.U, PRF_X.X, PRF_Y.Y,
              MKT_X.PX, MKT_Y.PY, MKT_L.W, MKT_K.R, MKT_U.PU,
              INC_I.I /;

*      Chose a numeraire: price of U fixed (.FX) at 1

PU.FX = 1;

*      Set initial values of variables (.L notation after variable)

X.L=100; Y.L=100; U.L = 200; I.L=200;
PX.L=1; PY.L=1; R.L=1; W.L=1; PU.L = 1;

SOLVE CLOSED USING MCP;

*      Counterfactual: double the endowment of labor
LBAR = 200;
SOLVE CLOSED USING MCP;

* Counterfactual: double the endowment of capital
LBAR = 100;
KBAR = 200;
SOLVE CLOSED USING MCP;

```

```

* Counterfactual: double the endowment of labor and capital
LBAR = 200;
KBAR = 200;
SOLVE CLOSED USING MCP;

*          Convert the model to a small open economy
*          Fix commodity prices and drop market clearing equations
*          for X and Y and for income balance.  Free up the wage rate.

MODEL SOE /PRF_U.U, PRF_X.X, PRF_Y.Y,
           MKT_L.W, MKT_K.R, MKT_U.PU, INC_I.I/;

*          Fix X and Y prices, unfix PU, set L back to benchmark = 100

PX.FX = 1; PY.FX = 1;
PU.UP = +INF; PU.LO = 0;
LBAR = 100; KBAR = 100;

SOLVE SOE USING MCP;

*          demonstrate the Rybszczyński theorem: double labor supply

LBAR = 200;
SOLVE SOE USING MCP;

*          demonstrate the Stolper-Samuelson theorem: increase px to 1.5

LBAR = 100;
PX.FX = 1.5;
SOLVE SOE USING MCP;

*          show a corner solution - economy specializes in X: px = 2

PX.FX = 2.0;
SOLVE SOE USING MCP;

```

Tables 1 and 2 show results for eight runs of the model. Table 1 gives results for the closed economy, Table 2 for a small open economy. The first column in each Table is the calibrated benchmark replication, where I have re-normalize all values to equal one initially. In Table 1, the second column gives results for doubling the labor endowment to $LBAR = 200$. Production shifts toward the labor intensive good Y , the relative price of Y falls, labor's wage falls relative to both commodity prices and the return to capital rises. This second column of Table 1 also illustrates what we could call aggregate diminishing returns. Doubling the endowment of one factor, which is fifty percent of income in the benchmark, increases total production of X and Y by less than fifty percent evaluated at initial prices. Welfare (not shown) increases from 1.0 in the benchmark to 1.41.

I often look at simple experiments where I know what the answer must be as yet another check on my model formulation and coding. Column three reverses the experiment of column 2, doubling the endowment of capital instead of labor. Because of the symmetry of X and Y in demand and because factor intensities are mirror images of each other in X and Y , the changes in the outputs of the two goods are the mirror images of those in column 2. Column 4 of Table 1 is a simple demonstration of the consequences of homogeneity of degree one in production and homothetic preferences. Doubling both factors of production just doubles both outputs and leaves all prices unchanged. Again, we know what the answer has to be so this is another check that the model formulation is correct.

Table 2 is intended to be close to that of the standard Heckscher-Ohlin model. Commodity prices are fixed at one to represent a small open economy (endowments are returned to their original level). The first column is the benchmark replication. The second column doubles the endowment of labor, results contrasting to Table 1 where prices change. Holding commodity prices constant, column 2 of Table 2 illustrated the Rybczynski theorem and Jones' (1965) magnification effect. Production of the labor intensive good more than doubles and the production of the capital intensive good shrinks. Unlike doubling labor in the closed economy, there are no aggregate diminishing returns: the added labor is absorbed without a fall in w by changing the composition of production.

Column 3 of Table 2 returns labor endowment to its original value and raises the relative (world) price of p_x to 1.5, a terms-of-trade improvement. X is capital intensive, and so the price increase results in the price of capital increasing by more than both commodity prices and the price of labor falls in terms of both commodity price. Regardless of consumption preferences, capital is better off and labor is worse off. This is the Stolper-Samuelson theorem.

I have included the last column of Table 2 to make another point about the limitations of local comparative statics that *assumes* an interior solution before and after a parameter change versus global analysis in a complementarity framework. If p_x is increased from its benchmark value of 1 to $p_x = 2$, the economy becomes specialized in X . Any further increase in p_x will continue to increase welfare, but it will have no effect on relative factor prices since all factors are employ in X . Stolper-Samuelson only works under the *assumption* of non-specialization.

5. Consistent modeling of a tax/tariff versus transport/trade cost in general equilibrium

Over the years, I have come across papers where there seem to be problems with the modeling of trade costs or tax distortions. Several different difficulties arise. The first is that income balance does not seem to hold in the model. A tax creates a wedge between world (or producer) price and consumer prices, generating revenue, but that revenue does not seem to appear (or reappear) elsewhere in the model. Or if the wedge between prices is a real trade/transport/transactions cost, then it must also be reflected elsewhere in the model to ensure supply/demand balance. A second issue is not a modeling error per se, but rather a modeling choice which misrepresents what the modeler is trying to do. This occurs when a wedge that is due to some real cost is modeled as an ad valorem tax/tariff with the revenue returned lump sum to consumers. Income balance will be satisfied, but the welfare effects of a tax and a real trade cost can be wildly different.

A third problem is modeling a quantitative restriction (e.g., a quota) as an ad valorem tax equivalent. Once one starts doing counterfactuals, the tax equivalent of the quota can change substantially, including going to zero. The correct procedure is to model the quota as an endogenous tax rate, with an added equation for the added unknown which adjusts the tax to maintain the quota level. But as with a tax, the quota rents (equivalent of tax revenue) must be allocated to some agent to maintain income balance in the model.

This section will tackle the first two problems. I will show how to correctly model both a tax and a real transactions cost in the simple model of the previous section. Then we will do a simulation to show that the welfare difference between the two is dramatic, emphasizing that real costs should not be treated as exogenous ad valorem tax wedges.

While an example from a model with international trade would be useful, let us stick with our current closed economy model and assume that there is a distortion between marginal cost and price in the X sector. One way to model this is as an ad valorem tax. The other is that there is some sort of real cost of moving the good from the producer to the consumer. A common and simple way of doing this is the so-called “iceberg” cost: some of the good melts and disappears in transit. This is in fact an ad valorem cost, since the share of the value lost as well as the quantity is constant.⁷ Let mc_x denote marginal cost as before, but also the producer price of X . Let TC denote the gross trade cost (one plus the ad valorem rate or power of the tax) between the producer and the consumer. p_x continues to denote the consumer price of X and X denotes the quantity *shipped*, so X/TC is the quantity *received* by the consumer. The model will have the following relationships:

⁷Trade economists value this assumption since it avoids having to explicitly model a transportation sector. But beware of modeling the “melt rate” as equal across sectors. That would mean that the same proportion of a shipment of microprocessors disappears in transit as that of a shipment of coal. Contrary to some accounts, the mathematical formulation referred to as iceberg trade cost is not due to Samuelson (1954), but to von Thünen (1842, 1863): a farmer takes grain to market in a horse-drawn cart and some of the grain must be fed to the horse. von Thünen provides actual empirical estimates of this cost! See Parr (2015).

$$\begin{aligned}
 mc_x TC &= p_x && \text{price paid by the importer} \\
 X/TC &&& \text{quantity received by the importer} \\
 mc_x X &= (mc_x TC)(X/TC) = p_x(X/TC) && \text{producer revenue = consumer cost}
 \end{aligned}$$

The last equation simply notes that the revenue received by producer equals what the consumer pays: the consumer pays a higher price per unit for a smaller quantity. There is no revenue floating around as there is with a tax.

Now also assume that there may be an ad valorem tax on X , and denote the gross tax (one plus the ad valorem rate) as TR . Assume arbitrarily that the tax base is inclusive of the transport/transactions cost. Then the pricing equation for the X sector corresponding to (21) and (40) above is

$$mc_x TC TR \geq p_x \quad \perp \quad X \quad (48)$$

Note that the transport cost and tax enter in the same way into the pricing equation. Perhaps in partial equilibrium, we could say they are equivalent. But they are not equivalent in general equilibrium as we will show shortly. Each wedge must appear somewhere else in the model. The transport cost will appear in the supply-demand balance equation for X , the equivalent of (25) and (44). The left-hand side must have the quantity received by the consumer.

$$X/TC \geq h_x U \quad \perp \quad p_x \quad (49)$$

The tax however appears somewhere else entirely in the model. Under the usual assumption, the tax revenue is returned lump sum to the representative consumer. Tax revenue is marginal cost times the gross transport cost rate times the *net tax rate* ($TR - 1$) times the quantity received by the consumer.

$$mc_x TC (TR - 1)(X/TC) = mc_x (TR - 1)X \quad \text{tax revenue} \quad (50)$$

So in the case of the tax, the consumer's budget balance equation equivalent to (28) and (47) becomes

$$I = wL + rK + mc_x (TR - 1)X \quad \perp \quad I \quad (51)$$

The gams code follows. The counterfactual experiment is to loop over a range of tax/transport cost rates. For each rate, the model is solved twice. First specifying that rates are tax rates, and secondly that the rates are transport costs. These are then plotted in Figure 1.

\$TITLE: JGEA2

* extends model JGEA1 by adding a tax and a (iceberg) transactions cost
 * and showing the substantial difference between the two

PARAMETERS

LBAR labor endowment
 KBAR capital endowment
 TC iceberg transportation (trade) cost on a gross basis (1+tc)
 TR tax on a gross basis (1+tr)
 WELFARE welfare;

LBAR = 100;
 KBAR = 100;
 TC = 1; TR = 1;

NONNEGATIVE VARIABLES

U activity level for utility or welfare
 X activity level for X production
 Y activity level for Y production

 W price of labor
 R price of capital
 PX price of good X
 PY price of good Y
 PU price of welfare (expenditure function)

 I income of the representative consumer;

EQUATIONS

PRF_U zero profit for welfare
 PRF_X zero profit for sector X
 PRF_Y zero profit for sector Y

MKT_L supply-demand balance for primary factor L
 MKT_K supply-demand balance for primary factor K
 MKT_X supply-demand balance for commodity X
 MKT_Y supply-demand balance for commodity Y
 MKT_U supply-demand balance for welfare

INC_I income balance;

* Zero profit inequalities

PRF_U.. $PX^{**0.5} * PY^{**0.5} =G= PU;$
 PRF_X.. $W^{**0.25} * R^{**0.75} * TC * TR =G= PX;$
 PRF_Y.. $W^{**0.75} * R^{**0.25} =G= PY;$

* Market clearance inequalities

MKT_L.. $LBAR =G= 0.25*(R/W)^{**0.75}*X +$
 $0.75*(R/W)^{**0.25}*Y;$
 MKT_K.. $KBAR =G= 0.75*(W/R)^{**0.25}*X +$
 $0.25*(W/R)^{**0.75}*Y;$
 MKT_X.. $X/TC =G= 0.5*U*PU/PX;$


```

MKT_Y..      Y      =G= 0.5*U*PU/PY;
MKT_U..      U      =G= I/PU;

*           Income balance equation

INC_I..      I =E= LBAR*W + KBAR*R + (TR-1)*(W**0.25*R**0.75)*X;

MODEL CLOSED /PRF_U.U, PRF_X.X, PRF_Y.Y,
              MKT_L.W, MKT_K.R, MKT_X.PX, MKT_Y.PY, MKT_U.PU,
              INC_I.I /;

*           Chose a numeraire: price of U fixed (.FX) at 1

PU.FX = 1;

*           Set initial values of variables (.L notation after variable)

X.L=100; Y.L=100; U.L = 200; I.L=200;
PX.L=1; PY.L=1; R.L=1; W.L=1; PU.L = 1;

SOLVE CLOSED USING MCP;

* SHOW HOW TO DO MULTIPLE SCENARIOS
* SHOW DIFFERENCE BETWEEN TARIFF AND TRADE COST OF EQUAL RATES

SETS T indexes 25 different gross cost levels          /T1*T25/
     J indexes 2 scenarios: 1 = tariff 2 = trade cost /J1*J2/;

PARAMETERS
  RATE(T)          net tax or trade cost rate (gross rate minus 1)
  WELFARE(T,J)     welfare normalized to equal 1 in benchmark
  RESULTS(T, *)    formats results in one table;

LOOP(J,
LOOP(T,

TC = 1; TR = 1;
RATE(T) = 1 + .05*ORD(T) - 0.05;
TR$(ORD(J) EQ 1) = RATE(T);
TC$(ORD(J) EQ 2) = RATE(T);

SOLVE CLOSED USING MCP;

WELFARE(T,J) = U.L;

);
);

RESULTS(T, "RATE") = RATE(T)-1;
RESULTS(T, "WELTR") = WELFARE(T, "J1")/200;
RESULTS(T, "WELTC") = WELFARE(T, "J2")/200;

DISPLAY RESULTS;

* Write parameter RESULTS to an Excel file JGEA.XLS,
* starting in Sheet1

Execute_Unload 'JGEA.gdx' RESULTS
execute 'gdxrw.exe JGEA.gdx par=RESULTS rng=SHEET1!'

```

The welfare results comparing the tax and trade/transport cost are shown in Figure 1,

where welfare has been normalized to equal 1 in the benchmark and the net cost rate is plotted on the horizontal axis. There are two things to note about the results. The first is that the welfare difference between the two is striking.

The second thing that stands out is that even a rather large tax rate has a rather small effect on welfare. The slope of the welfare curve at a zero tax is in fact zero. Readers familiar with the analysis of taxes and tariffs will recognize this result. The simplest way to see this is to think of extending the horizontal axis into the negative quadrant (a subsidy). Since there are no distortions in the model other than this tax or subsidy, the optimal value of the tax is zero. A zero tax rate is the peak of the welfare curve and that curve is flat at the maximum point of zero tax. The “first-order” effect of the tax is zero, and analytical models note that the distortion is multiplicative in the tax rate. This is not true with the transport charge: welfare falls steeply even in the neighborhood of a zero tax.

One important lesson to take from Figure 1 is the danger of taking something that is actually a real-resource-using distortion and modeling it as a tax/subsidy where the proceeds are returned lump-sum (or a subsidy raised lump-sum). The tax-cum-redistribution assumption may grossly underestimate the welfare loss of any distortion. I remember discussions decades ago with John Whalley and others, joking that the answer to any question put to a CGE model is “1.5 percent of GNP”. In most all cases, the models were examining taxes and subsidies with lump-sum redistribution, and perhaps Figure 1 gives some intuition as to why the effects are so small. Alternatively, consider for example a tax system in which the administrative costs are exactly equal to the tax collected. Those resources, such as inspectors, lawyers, accountants etc., have opportunity costs of actually working productively. In this case, the correct model formulates such a tax as a trade or transaction cost, giving the lower curve in Figure 1.

5. Consistent modeling of a price wedge (tax) versus a quantitative restriction (quota)

Another issue that deserves some comment concerns modeling price distortions such as a tax and quantitative distortions such as quotas or licenses. Modelers have often dealt with quotas or related quantitative restrictions (e.g., zoning rules on land use, pollution emission restrictions) by converting them to taxes. It is true that a benchmark quota restriction can generally be duplicated by a tax. But there are two issues that are quite important when the modeler performs counterfactuals.

The first occurs when the modeler continues to treat the quota as a fixed tax rate in those counterfactuals. This will inevitably give a wrong answer relative to holding the quantity fixed at the quota level. The second issue is that quotas or licenses create “quota rents”, which are equal to the revenues generated by the tariff equivalent at the benchmark. However, quota rents are very often distributed in different ways than tax revenues. Lucky folks or firms who are given licenses capture the rents, not the government. This is in fact an important source of corruption in many countries.

In this section, I will briefly show how to correctly model a quantitative restriction using our same template model. I will illustrate how a tax and quota, while the same in the benchmark,

become quite different as the economy grows. Incorporation alternative methods of distributing quota rents requires a more complete model and so will not be dealt with here.

Assume that the quota is a maximum constraint: values less than the quota are permitted and unconstrained. There are two ways of correctly modeling a quota. One is as an endogenous tax rate, which is set by an auxiliary equation that keeps the quota in force, though the quota may be slack and non-binding. The second way is to introduce an additional good, call it licenses. Then model a supply and demand for licenses, these licenses being a necessary input into one sector or trade activity. We will use the first method here (my teaching materials also model the license formulation).⁸

To maintain a close link to the previous examples, we assume that there is a maximum allowed production of good X . We will assume an endogenous tax rate QR which can be compared to the usual ad valorem tax rate TR . In this section, we model both on a net basis, which turned out to be easier in treating corner solutions. So the old TR value is now $(1 + TR)$. Here are the additions to the previous model.

PARAMETERS

XQUOTA maximum level of X allowed;

NONNEGATIVE VARIABLES

QR ad valorem tariff equivalent of the quota (net basis);

EQUATIONS

QCONS equation giving the quota inequality that sets QR;

QCONS.. XQUOTA =G= X;

Variable QR is complementary to the inequality $QCONS$. The pricing equations and the income balance equation are almost unchanged, just re-specified in net terms. These are:

PRF_X.. W**0.25 * R**0.75 * (1+QR) * (1+TR) =G= PX;

INC_I.. I =E= LBAR*W + KBAR*R + (TR) * (W**0.25*R**0.75) *X
+ (QR) * (W**0.25*R**0.75) *X;

In this simple model, the quota rents are redistributed just like tax revenues. The only remaining thing is to add the added equation and unknown to the model declaration statement. The .gms files for this version is JGAE3.

We solve our model with a tax rate of 0.25. The solution value for X at this tax is then used to set the quota parameter $XQUOTA$, which is then held fixed. The experiment is then to shrink and expand the size of the economy from its benchmark value (same size as in the previous examples). Benchmark level of welfare is the same for the tax and quota, but they will diverge as the economy changes size. The model is first solved with QR fixed at zero ($TR = 0.25$), and then solved with QR set endogenously with $TR = 0$, for each level of size.

⁸A quite different approach to quotas and other inequalities in GEMPACK is found in Harrison, Horridge, Pearson and Wittwer (2004). Brief comments are found in the appendix below.

Results are shown in Figure 2. The vertical dashed line is the benchmark size of the economy. The left-hand vertical axis plots the ratio of welfare under the quota to that under the tax. This ratio is of course equal to one at the benchmark. The right-hand vertical axis plots the quota equivalent tax QR that holds X at its maximum value. If the quota is non-binding, then QR is zero.

Figure 2 shows that, if we shrink the economy by moving left from the dashed line, QR quickly falls to zero (quota is non-binding) and welfare rises a little from its benchmark value of one. The latter result naturally follows from the fact that there are no other distortions in this competitive model, and so any tax or binding quota reduces welfare from its maximum value. Figure 2 shows that expanding the size of the economy from its benchmark level requires a large increase in the tax rate QR in order to preserve X production at its benchmark value. The Figure also shows that the welfare ratio falls significantly moving to the right (welfare under the tax increases linearly with the economy's size).

The message here is rather clear. First, quantitative restrictions should not be modeled by a fixed price wedge which is then held constant in counter-factual analysis. Second, it is not very hard to model a quantitative restriction as an endogenous tax rate. It requires only a fairly simple modification of the model. Also, our complementarity formulation allows the quantitative restriction to be non-binding, an important feature in many real world cases (the European emissions trading scheme?). Modeling a quantitative restriction by adding a good called licenses is also easy, and allows for the modeling of corruption through alternative methods of allocating the licenses.

5. Summary

Trade economists made fundamental contributions to general equilibrium analysis by formulating models using the building blocks of what we now call duality techniques. These produced models which were far more useful for the analysis of practical questions of the type asked by trade and public-economics economists than earlier analyses focusing on existence, uniqueness, and stability of equilibria. Local comparative statics analysis was used to ask questions about changing factor endowments, changing technologies, changing world prices and changing trade and domestic taxes. This immensely improved our ability to understand such things as the relationship between world commodity prices and domestic income distribution.

Limitations remained of course. The analysis was for small changes only, results were qualitative (signs and some relative magnitudes) and the method was generally restricted to interior solutions only in which initially positive variables could not go to zero or vice versa. What this paper shows however, is that the use of duality tools such as converting production functions and utility functions to cost and expenditure functions paved the way for a more complete global analysis using complementarity built on the foundations of the Karush-Kuhn-Tucker theorem. I show how key tools and theorems lead naturally to a formulation that allows large changes, yields quantitative results needed by policy makers, and allows corner solutions to emerge or disappear in response to changing parameters such as technologies, trade costs or tariffs.

Specific functional forms are needed and indeed specific parameter values for those functions. But specific functional forms are always needed if one wants quantitative results. In models with scale economies and imperfect competition, even qualitative results cannot be obtained without specific functional forms. Often parameters can be drawn from literature estimates or estimated econometrically as part of the analysis at hand. Sensitivity analysis can indicate which parameters have major or minor effects on the results. But global simulation analysis has indeed improved our ability to provide some answers to important public policy questions.

The final two sections of the paper examine two situations which illustrate the need from global analysis or alternatively while some standard practices are not appropriate for large changes. In the first, I illustrate how different a tax with lump-sum redistribution is from a real resource using (e.g., iceberg) trade or transaction cost. The former imposes very small welfare losses, perhaps accounting for many of the very small effects in CGE models. The latter effects are much larger.

The second example illustrated how a quantitative restriction such as a quota or license scheme can lead to very different effects from a price distortion such as a tax. While the two can be calibrated to yield the same results at some benchmark, significant parameter changes in counter-factual experiments produce very different effects depending on whether a fixed tax is maintained or a fix quota is maintained instead.

REFERENCES

- Auspitz, Rudolf and Richard Lieben (1889), *Untersuchungen über die Theorie des Preises*, Leipzig: Duncker & Humblot.
- Codsi, G. and K. R. Pearson (1988), “GEMPACK: General-Purpose Software for Applied General Equilibrium and Other Economic Modellers”, *Computer Science in Economics and Management* 1, 1889-207.
- Corong, Erwin L., Thomas W. Hertel, Robert A. McDougall, Marinos E. Tsigas and Dominique van der Mensbrugge (2017), “The standard GTAP Model, Version 7”, *Journal of Global Economic Analysis* 2(1), 1-119.
- Cox, David and Richard G. Harris (1985), “Trade Liberalization and Industrial Organization - Some Estimates for Canada”, *Journal of Political Economy* 93(1), 115-145.
- Dixit, Avinash and Victor Norman (1980), *Theory of International Trade*, Cambridge University Press.
- Dixon, P.B and M. T. Rimmer (2016), “Johansen’s legacy to CGE modelling: originator and guiding light for 50 years”, *Journal of Policy Modeling* 38(3), 421-35.
- Dixon, P.B., B.R. Parmenter, John Sutton and D.P. Vincent (1982), ORANI: “A Multisectoral Model of the Australian Economy”, *Contributions to Economic Analysis 142*, North-Holland Publishing Company, xviii + 372.
- Harris, Richard G. (1984), “Applied General Equilibrium Analysis of Small Open Economies with Scale Economies and Imperfect Competition”, *American Economic Review* 734(5), 1016-1032.
- Harrison, Glen W., Thomas F. Rutherford, and Ian Wooton (1989), “The Economic Impact of the European Community”, *American Economic Review* 79(2), 288-294.
- Harrison, W.J., M. Horridge, K. Pearson and G. Wittwer (2004), “A Practical Method for Explicitly Modeling Quotas and Other Complementarities”, *Computational Economics* 23, pp.325-341. <https://doi.org/10.1023/B:CSEM.0000026795.44450.4a>
- Hotelling, Harold (1932), “Edgeworth’s Taxation Paradox and the Nature of Demand and Supply Function”, *Journal of Political Economy* 40, 577-616.

<http://spot.colorado.edu/~markusen> see teaching section for full course materials.

<http://www.copsmodels.com/archivep.htm#tpmh0191>

- IBRD (1981), "General Algebraic Modeling System: Preliminary User's Guide", Washington D.C.: The World Bank, Development Research Center.
- Johansen, Leif (1960), "Rules of Thumb for the Expansion of Industries in a Process of Economic Growth", *Econometrica* 28(2), 258-271.
- Jones, Ronald W. (1956), "Factor Proportions and the Heckscher-Ohlin Theorem", *Review of Economic Studies* 24, 1-10.
- Jones, Ronald W. (1965), "The Structure of Simple General Equilibrium Models," *Journal of Political Economy* 73, 557-572.
- Karush, William (1939), *Minima of Functions of Several Variables with Inequalities as Side Constraints*, M.Sc. thesis, Department of Mathematics, University of Chicago.
- Kuhn, Harold W. and Albert W. Tucker (1951), "Nonlinear Programming", *Proceedings of 2nd Berkeley Symposium*. Berkeley: University of California Press, 481-492.
- Mathiesen, Lars (1985), "Computation of Economic Equilibria by a Sequence of Linear Complementarity Problems", *Mathematical Programming* 23, 144-162.
- Markusen, James R. (2002), *Multinational Firms and the Theory of International Trade*, Cambridge: MIT Press.
- Markusen, James R. and Anthony J. Venables (2007), "Interacting Factor Endowments and Trade Costs: A Multi-Country, Multi-Good Approach to Trade Theory", *Journal of International Economics* 73, 333-354.
- Markusen, James R. (2013), "Expansion of Trade at the Extensive Margin: A General Gains-from-Trade Result and Illustrative Examples", *Journal of International Economics* 88, 262-270.
- Markusen, James R. (2021), "Ronald Jones' Duality Analysis as a Foundation for Applied General-Equilibrium Modeling", *International Journal of Economic Theory* 17 (2021), 6-19.
- Meeraus, Alexander (1983), "An Algebraic Approach to Modeling", *Journal of Economic Dynamics and Control* 5, 81-108.
- Parr, John B. (2015), "Overlooked Aspects of the von Thünen System", *Spatial Economics* 10, 471-487.
- Roy, René (1947), "La Distribution du Revenu Entre Les Divers Biens", *Econometrica* 15(3): 205-225.
- Rutherford, Thomas F. (1985), *MPS/GE user's guide*, Department of Operations Research, Stanford University.

- Rutherford, Thomas F. (1995), "Extensions of GAMS for complementarity problems arising in applied economic analysis", *Journal of Economic Dynamics and Control* 19(8), 1299-1324.
- Samuelson, Paul (1954). "The Transfer Problem and Transport Costs, II: Analysis of Effects of Trade Impediments", *Economic Journal* 64, 264–289.
- Scarf, Herbert (1967), "Approximation of Fixed Points of a Continuous Mapping", *SIAM Journal of Applied Mathematics* 15(5), 1328-1343.
- Schmidt, Torsten (2004), "Really Pushing the Envelope: Early Use of the Envelope Theorem by Auspitz and Lieben", *History of Political Economy* 36, 103-129.
- Shepard, Ronald (1953), *The Theory of Cost and Production Functions*, Princeton: Princeton University Press.
- Shoven, John B. and John Whalley (1973), "General Equilibrium with Taxes: Computational Procedure and an Existence Proof", *Review of Economic Studies* 40(4), 475-489.
- Shoven, John B. and John Whalley (1974), "Computation of Competitive Equilibrium on International Markets with Tariffs", *Journal of International Economics* 4(4), 341-354.
- Takayama, T. and Alan D. Woodland (1970), "Equivalence of Price and Quantity Formulations of Spatial Equilibrium: Purified Duality in Quadratic and Concave Programming", *Econometrica* 38(6), 889-906.
- Woodland, Alan D. (1980), "Direct and Indirect Trade Utility Functions", *Review of Economic Studies* 47(5), 907-926,
- Woodland Alan D. (1982), *International Trade and Resource Allocation*, Amsterdam: North Holland.
- Von Thünen, J. H. (1842), *Der isolirte Staat in Beziehung auf Landwirthschaft und Nationalökonomie*, Part One, Rostock: Leopold.

Appendix:

This article is about how the traditional tools taught in microeconomic theory courses and used in local analysis can indeed be exploited to create GE models that allow the analysis of large changes and permit activities to switch from active to inactive or vice versa. Both of these features are fundamental to my work on endogenous multinational firms and location choices. The objection is to show how a theory-consistent model is formulated and operationalized.

It is not my intention to write a history of general-equilibrium modeling and especially not to analyze the merits of alternative software approaches and solution algorithms. Those items are not within my skill set. Nevertheless, I think it may be helpful to offer my incomplete and superficial review as a guide to those who might like to look further.

I suggest that three approaches to AGE analysis were developed in parallel during the 1970s and 1980s, though theoretical foundations go back much further (I will not engage in a food fight about what or who was first). One approach was initiated by the algorithm of Scarf (1967) with the first large-scale implementations by Shoven and Whalley (1973, 1974). This algorithm and its refinements used an iterative fixed-point procedure to solve high-dimension models. But there were a number of inherent limitations in this approach, including slow computation and the inability to handle crucial extensions to basic competitive models such as scale economies, imperfect competition, public goods, externalities and so forth.

A second (in no particular order) approach is associated with a group of Australian economists and the ORANI project and the GEMPACK software. I am relying largely on the report of one referee here and this referee indicates that the GTAP software is in turn descendent from the Australia contributions. A good introduction to GTAP with a bit more history of thought is found in Corong et. al. (2017). I have no personal experience using these softwares but believe I understand the basics.

A GEMPACK reference is Dixon, Parmeter, Sutton and Vincent (1982). Their approach builds on earlier work of Johansen (1960) and a good reference that explains the methodology may be Dixon and Rimmer (2016). A key aspect of this is to first linearize the set of non-linear equation and then solve the linearized version. A good exposition of this is found in Codsí and Pearson (1988), but of course many improvements were made subsequently. Two limitations of the initial approach in my view were that (1) calculations based on the linearized system are only locally valid for small changes and (2) strictly interior solutions to comparative statics had to be assumed, ruling out a great many issues. As I understand it, the first issue was solved by updating parameters such as value and employment shares (my a_{ij} for example) and re-solving the linear system. Then repeat. The second can be handled by extensions as shown in Harrison, Horridge, Pearson and Wittwer (2004). My own opinion, surely not shared by this group, is that this method for allowing regime switches is awkward, hard to follow, and may not be scalable to large numbers of non-linear inequalities.

A third approach is that of GAMS (IRBD 1981, Meeraus (1983)), which has strong foundations in the KKT theorem and other theoretical results noted earlier. Rutherford (1985) independently took a similar approach following Mathiesen (1985), with his MPS/GE software, later incorporated into GAMS. This approach, non-linear complementarity as described above, allows one-shot solutions to large-change comparative statics and permits regime switching and

corner solutions. Rutherford's MPS/GE allows for a much simplified benchmark calibration procedure, using exactly the data that the modeler has initially. A more complete guide is found in Rutherford (1995) and an early example of its application is Harrison, Rutherford and Wooton (1989). I have used the non-linear complementarity approach extensively in my own work in which endogenous regime switching is of fundamental interest (Markusen 2002).

Finally, I credit Harris (1984) and Cox and Harris (1985) with important breakthroughs, especially for small economies, in incorporating scale economies and imperfect competition into AGE models. These extensions or rather departures from the traditional competitive models showed much larger costs of protection for smaller countries and made a major contribution to Canadian politics in the changing attitude toward free trade.

I end with my explanation of what is meant by "linearization", and then expressing the linearized version in proportional change terms. For this, I will use just one equation from the model above, which is the zero profit condition for the X sector. As per the earlier exposition (ignoring the weak inequality) the equation and the result from Shepard's lemma is:

$$c_x(w, r) = p_x \quad \frac{\partial c_x}{\partial w} = a_{lx}(w, r) \quad \frac{\partial c_x}{\partial r} = a_{kx}(w, r) \quad (\text{A1})$$

Since unit cost is homogeneous of degree 1 in prices, the pricing equation for X is then

$$c_x(w, r) = a_{lx}(w, r)w + a_{kx}(w, r)r = p_x \quad (\text{A2})$$

Totally differentiating this equation gives

$$a_{lx}dw + a_{kx}dr + [wda_{lx} + rda_{kx}] = a_{lx}dw + a_{kx}dr = dp_x \quad (\text{A3})$$

But the term in square brackets is locally zero as a consequence of the envelop theorem. The slope of an isoquant is the ratio of the factor prices at the initial equilibrium.

$$\frac{w}{r} = -\frac{da_{kx}}{da_{lx}} \quad \Rightarrow \quad [wda_{lx} + rda_{kx}] = 0 \quad (\text{A4})$$

Multiply (A3) through by p_x . Multiple and divide the first term by w , and the second term by r .

$$\left[\frac{a_{lx}w}{p_x} \right] \frac{dw}{w} + \left[\frac{a_{kx}r}{p_x} \right] \frac{dr}{r} = \frac{dp_x}{p_x} \equiv \hat{p}_x \quad (\text{A5})$$

The terms in square brackets are, respectively, the value share of labor and the value share of capital in X . Denoting these value shares with thetas (original Jones notation I recall), equation (A5) is given in proportional changes as

$$\theta_{lx} \hat{w} + \theta_{kx} \hat{r} = \hat{p}_x \quad (\text{A6})$$

Repeating this procedure for all equations in the model, gives a linear general equilibrium model where the thetas and similar variables are treated as parameters and the hat terms are (generally) the variables. It can then be solve as a linear system for "shocks" such as changing world prices,

technology or trade costs. I think that this is what GEMPACK does as an initial step. It is only locally valid since, as emphasized in (A3), changes in the thetas locally net to zero. As I understand it, what GEMPACK does then is to update the thetas and similar parameters (since they are functions of other endogenous variables) based on the linear solution. Then the same linear problem is solved again. Updating repeated, solved again until there is convergence.

One referee asked that I discuss how to recognize my template (non-linear complementarity) or its absence in other modeling formats. I honestly do not have the experience with using these alternatives to do so and, as I have indicated too many times, this paper is not about a comparison and evaluation of alternative software. But I can offer some comments that could help a new user get started. Much depends on the issues and economies that you wish to analyze. If you want model multi-region economies with perfect competition, constant returns to scale, and Armington product differentiation (goods are differentiated by country of origin), then all the options I have listed are fine. The Armington assumption in trade models basically guarantees that there is always an interior solution to comparative statics experiments so complementarity is not needed. But collectively, these assumption almost guarantee that trade policy changes will have quite small effects on welfare.

If one wants to incorporate richer features such as increasing returns to scale, endogenous oligopolistic competition and markup formulas, regime switching and expansion of trade at the extensive margin, externalities and public goods, solve for optimal taxes and so forth, then the choice narrow. I will not claim that alternative software cannot do many of these things, since I am not certain. But I will claim that the non-linear complementarity approach can incorporate all of them.

Table 1: Closed economy results (variables normalized to equal 1 in benchmark)

Variable	Benchmark	Double labor Endowment	Double capital Endowment	Double both labor and capital
U	1.00	1.41	1.41	2.00
X	1.00	1.19	1.68	2.00
Y	1.00	1.68	1.19	2.00
PX	1.00	1.19	0.84	1.00
PY	1.00	0.84	1.19	1.00
W	1.00	0.71	1.41	1.00
R	1.00	1.41	0.71	1.00
PU	1.00	1.00	1.00	1.00
I	1.00	1.41	1.41	2.00

Notes to Table 1: Column 2 illustrates aggregate diminishing returns
 Column 3 illustrates the symmetry in the model
 Column 4 illustrates homogeneity of the economy under constant returns in production and homothetic preferences

Table 2: Small open economy results

Variable	Benchmark	Double labor Endowment	Increase price of good X by 50%	Increase price of good X by 100%
U	1.00	1.50	1.08	1.24
X	1.00	0.50	1.56	1.75
Y	1.00	2.50	0.31	
PX	1.00	1.00	1.50	2.00
PY	1.00	1.00	1.00	1.00
W	1.00	1.00	0.82	0.88
R	1.00	1.00	1.84	2.63
PU	1.00	1.00	1.23	1.41
I	1.00	1.50	1.33	1.75

Notes to table 2: Column 2 illustrates the Rybczynski Theorem
 Column 3 illustrates the Stolper-Samuelson Theorem
 Column 4 illustrates the importance of not ruling out corner solutions (specialization)

Figure 1: Comparison of tax and transactions cost




Figure 2: Non-equivalence of tax and quota as economy grows

